

Lesson 12

Integration using Partial Fractions

Initializations

```
> restart;  
with(student):  
with(oneonta):
```

12.1 Partial Fraction Decomposition

The Maple command used to expand a rational function into partial fractions is given by:
convert(expression, parfrac, variable);

Examples

Example 12.1.1

Split into partial fractions $\frac{5x + 7}{(x-2)(x+3)}$.

Solution

Maple has a facility which converts a rational function into partial fractions. The required syntax is listed below.

```
> e1 := (5*x+7)/((x-2)*(x+3));
```

$$e1 := \frac{5x + 7}{(x - 2)(x + 3)} \quad (2.1.1.1)$$

```
> e2 := convert(e1, parfrac, x);
```

$$e2 := \frac{8}{5(x + 3)} + \frac{17}{5(x - 2)} \quad (2.1.1.2)$$

```
>
```

Example 12.1.2

Evaluate $\int \frac{3x + 4}{(x^2 + 3)(5 - 2x)} dx$.

Solution

The power of a partial fraction decomposition lies in its usefulness as an integration tool. The **parf** routine in the **oneonta** package allows the user to perform a partial fraction decomposition inside an integral expression.

```
> e1 := Int((3*x+4)/((x^2+3)*(5-2*x)), x);
```

$$e1 := \int \frac{3x + 4}{(x^2 + 3)(5 - 2x)} dx \quad (2.1.2.1)$$

```
> e2:=parf(e1);
```

$$e2 := \int \left(\frac{1}{37} \frac{23x+2}{x^2+3} - \frac{46}{37(-5+2x)} \right) dx \quad (2.1.2.2)$$

Let Maple evaluate these elementary integrals.

```
> e3:=value(e2)+C;
```

$$e3 := \frac{23}{74} \ln(x^2+3) + \frac{2}{111} \sqrt{3} \arctan\left(\frac{1}{3} x \sqrt{3}\right) - \frac{23}{37} \ln(-5+2x) + C \quad (2.1.2.3)$$

```
>
```

Example 12.1.3

Evaluate $\int \frac{1}{(x^2+5)^2} dx$.

Solution

Even though it may be tempting to apply partial fraction decomposition to this integral, close examination reveals that the integrand is already in decomposed form. A problem like this is solved using a trigonometric substitution.

```
> e1:=Int(1/(x^2+5)^2, x);
```

$$e1 := \int \frac{1}{(x^2+5)^2} dx \quad (2.1.3.1)$$

Use the substitution $x = \sqrt{5} \tan u$. Immediately compute the inverse transformation, because we will need it at the end of the computation to express our result in terms of the variable x .

```
> rr:=x=sqrt(5)*tan(u);
```

```
inv_rr:=isolate(rr, u);
```

$$rr := x = \sqrt{5} \tan(u)$$

$$inv_rr := u = \arctan\left(\frac{1}{5} x \sqrt{5}\right) \quad (2.1.3.2)$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int \frac{\sqrt{5}}{25 + 25 \tan(u)^2} du \quad (2.1.3.3)$$

```
> e3:=simplify(e2);
```

$$e3 := \frac{1}{25} \sqrt{5} \left(\int \cos(u)^2 du \right) \quad (2.1.3.4)$$

Of course this integral can be evaluated using a double angle formula. We simply ask for its value.

```
> e4:=value(e3)+C;
```

$$e4 := \frac{1}{25} \sqrt{5} \left(\frac{1}{2} \cos(u) \sin(u) + \frac{1}{2} u \right) + C \quad (2.1.3.5)$$

```
> e5:=subs(inv_rr, e4);
```

$$e5 := \frac{1}{25} \sqrt{5} \left(\frac{1}{2} \cos\left(\arctan\left(\frac{1}{5} x \sqrt{5}\right)\right) \sin\left(\arctan\left(\frac{1}{5} x \sqrt{5}\right)\right) \right) \quad (2.1.3.6)$$

$$+ \frac{1}{2} \arctan\left(\frac{1}{5} x \sqrt{5}\right) + C$$

```
> e6:=expand(e5);
```

$$e6 := \frac{1}{2} \frac{x}{5x^2 + 25} + \frac{1}{50} \sqrt{5} \arctan\left(\frac{1}{5} x \sqrt{5}\right) + C \quad (2.1.3.7)$$

The result can be simplified slightly by mapping the **simplify** routine to the expression above.

```
> ans:=map(simplify, e6);
```

$$ans := \frac{1}{10} \frac{x}{x^2 + 5} + \frac{1}{50} \sqrt{5} \arctan\left(\frac{1}{5} x \sqrt{5}\right) + C \quad (2.1.3.8)$$

```
>
```